Je viens de récupérer un fichier script.rpy d'un jeu. Te semble-t-il complet ? 


default persistent.chance_fantasy = 0.25
define transition_time = 0.5



label common_start_level:
    $ temp_level_number += 1
    $ level_number = 0
    pause 0.1
    $ level_number = temp_level_number
    $ start_level_setup()
    show screen block
    show screen gameplay
    $ renpy.pause(0.5)
    if len(active_deck) > 0:
        call draw from _call_draw
        $ renpy.pause(0.4)
    if len(active_deck) > 0:
        call draw from _call_draw_1
        $ renpy.pause(0.4)
    if len(active_deck) > 0:
        call draw from _call_draw_2
        $ renpy.pause(0.5)
    hide screen block
    return


label start:
    jump expression "level_1_" + persistent.selected_girl



label level_1_Miriam:
    scene university_bg with Dissolve (transition_time)
    $ renpy.pause(0.5)
    show lv1_particles

    show screen player_interface with Dissolve (transition_time)
    $ save_game("last_save")

    #voice "voices/mc/mc_lv0.ogg"
    mc "It's the start of a new day... time to get dressed and go to University."

    $ basic_shirt.choose_clothes()
    $ basic_pants.choose_clothes()
    $ basic_shoes.choose_clothes()







    $ start_level_setup()
    $ renpy.pause(0.5)

    $ set_level_images("lv1")
    $ nr = str(renpy.random.randint(1, 4))
    call expression "d_lv1_" + nr + "_" + persistent.selected_girl from _call_expression

    $ cards_to_choose = choose_cards(hand_lv1)
    call screen get_a_card(cards_to_choose)
    $ renpy.pause(0.3)

    $ cards_to_choose = choose_cards(hand_lv1)
    call screen get_a_card(cards_to_choose)
    $ renpy.pause(0.3)

    $ cards_to_choose = choose_cards(hand_lv1)
    call screen get_a_card(cards_to_choose)
    $ renpy.pause(1)

    $ stage_number = 1
    call common_start_level from _call_common_start_level
    $ random_list = [commonmate, friend, john, louis, rich_dude]
    $ sure_list = [deskjob, basic_store, basic_hobby]
    $ level_list = set_level_list(random_list, sure_list, 5)
    $ end_level_setup()
    jump new_turn


label level_2_Miriam:
    $ persistent.level_cleared.add("lv1_Miriam")
    $ set_level_images("lv1")
    show screen player_interface with Dissolve (transition_time)
    $ nr = str(renpy.random.randint(1, 4))
    call expression "d_lv2_" + nr + "_" + persistent.selected_girl from _call_expression_1

    $ cards_to_choose = choose_cards(hand_lv2)
    call screen get_a_card(cards_to_choose)
    $ renpy.pause(0.1)

    call common_start_level from _call_common_start_level_1
    $ level_list = [commonmate, friend, john, louis, rich_dude, deskjob, basic_present, basic_store_active, basic_store_boost, basic_store_control, basic_hobby, wingman, meet_her]
    if len(configure_fantasy_list(1, 4)) > 0:
        if renpy.random.random() < persistent.chance_fantasy:
            $ level_list.append(fantasy_campus)
    $ end_level_setup()
    jump new_turn

label level_3_Miriam:
    $ persistent.level_cleared.add("lv2_Miriam")
    $ set_level_images("lv3")
    show screen player_interface with Dissolve (transition_time)
    $ nr = str(renpy.random.randint(1, 4))
    call expression "d_lv3_" + nr + "_" + persistent.selected_girl from _call_expression_2

    $ cards_to_choose = choose_cards(hand_lv3)
    call screen get_a_card(cards_to_choose)
    $ renpy.pause(0.1)

    call common_start_level from _call_common_start_level_2

    $ stage1_stores = [basic_store_active, basic_store_boost, basic_store_control]
    $ renpy.random.shuffle(stage1_stores)
    $ random_list = [commonmate, friend, john, louis, rich_dude, wingman]
    $ sure_list = [deskjob, basic_store, basic_hobby, basic_present, basic_present2, political_talk, meet_girlfriends, shy_classmate, calculus_professor, quick_rival, talk_her, taunt_rival] + [stage1_stores[0]]
    $ level_list = set_level_list(random_list, sure_list, 4)
    if len(configure_fantasy_list(1, 4)) > 0:
        if renpy.random.random() < persistent.chance_fantasy:
            $ level_list.append(fantasy_campus)
    $ end_level_setup()
    jump new_turn

label level_4_Miriam:
    $ persistent.level_cleared.add("lv3_Miriam")
    $ set_level_images("lv4")
    show screen player_interface with Dissolve (transition_time)
    $ nr = str(renpy.random.randint(1, 4))
    call expression "d_lv4_" + nr + "_" + persistent.selected_girl from _call_expression_3

    $ cards_to_choose = choose_cards(hand_lv4)
    call screen get_a_card(cards_to_choose)
    $ renpy.pause(0.1)

    call common_start_level from _call_common_start_level_3
    $ random_list = [political_talk, meet_girlfriends, shy_classmate, wingman, taunt_rival, aprof]
    $ sure_list = [deskjob, basic_hobby, basic_present, basic_present2, assflirt_her, religion_talk, hot_card, cute_card, coldbood_rival, cockblocker] + [stage1_stores[1], stage1_stores[2]]
    $ level_list = set_level_list(random_list, sure_list, 4)
    if len(configure_fantasy_list(1, 4)) > 0:
        if renpy.random.random() < persistent.chance_fantasy:
            $ level_list.append(fantasy_campus)
    $ end_level_setup()
    jump new_turn

label level_5_Miriam:
    $ persistent.level_cleared.add("lv4_Miriam")
    $ set_level_images("lv4")
    show screen player_interface with Dissolve (transition_time)

    scene expression "images/girls/[persistent.selected_girl]/lv4c/lv4c_1.png" onlayer forward at center_images_t with Dissolve (1.0)

    $ persistent.has_seen_girl_image.add(f"images/girls/{persistent.selected_girl}/lv4c/lv4c_1.png")
    $ nr = str(renpy.random.randint(1, 4))
    call expression "d_lv5_" + nr + "_" + persistent.selected_girl from _call_expression_4

    $ renpy.pause(0.3)
    "*First date confirmed*"
    "Stage 1 cleared!"
    hide expression "images/girls/[persistent.selected_girl]/lv4c/lv4c_1.png" onlayer forward with Dissolve (1.0)
    $ persistent.times_stage1_clear_Miriam += 1
    $ renpy.pause(0.3)


    call upgrade_card_label from _call_upgrade_card_label


    $ cards_to_choose = choose_cards(hand_lv5)
    call screen get_a_card(cards_to_choose)
    $ renpy.pause(0.1)

    hide screen player_interface with dissolve
    scene park_bg with Dissolve (1.0)
    $ stage_number = 0
    $ level_number = 0
    pause 0.1
    $ stage_number = 2
    $ current_girl_image = ""
    $ renpy.pause(0.3)
    $ set_level_images("lv5")
    show screen player_interface with dissolve
    $ renpy.pause(0.3)
    show lv5_particles

    call common_start_level from _call_common_start_level_4
    $ random_list = []
    $ sure_list = [team_player_card, team_captain_card, team_strong_card, team_fast_card, girlask_park_card, girlwork_park_card, girlpassing_park_card, girlswim_park_card,\
        coachjob, park_store, basic_rose, park_hobby, exercise_her, run_act]
    $ level_list = set_level_list(random_list, sure_list, 0)
    if len(configure_fantasy_list(5, 6)) > 0:
        if renpy.random.random() < persistent.chance_fantasy:
            $ level_list.append(fantasy_park)
    $ end_level_setup()
    $ level_number = 5

    jump new_turn

label level_6_Miriam:
    $ persistent.level_cleared.add("lv5_Miriam")
    $ set_level_images("lv6")
    show screen player_interface with Dissolve (1.0)
    $ nr = str(renpy.random.randint(1, 4))
    call expression "d_lv6_" + nr + "_" + persistent.selected_girl from _call_expression_5

    $ cards_to_choose = choose_cards(hand_lv5)
    call screen get_a_card(cards_to_choose)
    $ renpy.pause(0.1)

    call common_start_level from _call_common_start_level_5
    $ random_list = [girlask_park_card, girlwork_park_card, girlpassing_park_card, girlswim_park_card]
    $ sure_list = [smug_park_card, shirtless_park_card, officer_park_card, work_park_talk, hobbies_park_talk, chat_park_act, confess_her,\
        coachjob, park_store, basic_present, basic_rose, park_hobby, yoga_park_card]
    $ level_list = set_level_list(random_list, sure_list, 3)
    if len(configure_fantasy_list(5, 6)) > 0:
        if renpy.random.random() < persistent.chance_fantasy:
            $ level_list.append(fantasy_park)
    $ end_level_setup()

    jump new_turn

label level_7_Miriam:
    $ persistent.level_cleared.add("lv6_Miriam")
    $ set_level_images("lv6")
    show screen player_interface with Dissolve (1.0)

    scene expression "images/girls/[persistent.selected_girl]/lv6c/lv6c_1.png" onlayer forward at center_images_t with Dissolve (1.0)

    voice "voices/" + persistent.selected_girl + "/mer_lv7.ogg"
    g "I... I like you too"

    scene expression "images/girls/[persistent.selected_girl]/lv6c/lv6c_2.png" onlayer forward at center_images_t with Dissolve (1.0)
    $ persistent.has_seen_girl_image.add(f"images/girls/{persistent.selected_girl}/lv6c/lv6c_2.png")
    $ renpy.pause(0.5)
    "*First kiss*"

    scene expression "images/girls/[persistent.selected_girl]/lv6c/lv6c_1.png" onlayer forward at center_images_t with Dissolve (1.0)

    "Stage 2 cleared!"
    $ persistent.times_stage2_clear_Miriam += 1
    $ renpy.pause(0.5)
    hide expression "images/girls/[persistent.selected_girl]/lv6c/lv6c_1.png" onlayer forward with Dissolve (1.0)

    if persistent.times_stage2_clear_Miriam > 0:
        if "Norah" not in persistent.list_of_girls:
            $ persistent.list_of_girls.append("Norah")
            $ persistent.selected_girl = "Norah"
            $ set_level_images("lv1")
            play sound renpy.random.choice(sfx_win)
            "Congratulations! You unlocked a new girl to play with, Norah!"

    window hide
    $ renpy.pause(0.5)

    nvl show dissolve
    n "This is the end of the current version. Thanks for playing!"
    $ choice_dialogue_line = renpy.random.choice([0,1,2,3])
    if choice_dialogue_line == 0:
        n "Have you checked your rewards in the Main Menu?"
        n "The more you play, the more rewards you'll unlock!"
    elif choice_dialogue_line == 1:
        n "Please consider visiting my {a=https://mrsilverlust.itch.io/}Itch.io{/a}, {a=https://www.patreon.com/mrsilverlust/}Patreon{/a} or {a=https://discord.gg/hQNWkAc9Sm}Discord server{/a} to give your feedback."
        n "This game doesn't take any analytics and never connects to the Internet. Your feedback is the only thing I can use to balance cards, strategies and the dificulty curve."
    elif choice_dialogue_line == 2:
        n "For more news about future updates, please visit my {a=https://mrsilverlust.itch.io/}Itch.io{/a}, {a=https://www.patreon.com/mrsilverlust/}Patreon{/a} or {a=https://discord.gg/hQNWkAc9Sm}Discord server!{/a}"
        n "Your feedback and support is very appreciated!"
    elif True:
        n "If you liked the game, please consider leaving a review!"
        n "Your feedback and support is very appreciated!"
    nvl hide dissolve

    $ persistent.chance_fantasy = 0.25

    $ renpy.pause(0.5)
    hide screen player_interface with Dissolve (1.0)
    pause 0.5
    scene main_menu with Dissolve (1.5)
    pause 0.5
    $ renpy.unlink_save("last_save")
    $ renpy.full_restart()







label exp_none_0:
    return

label exp_present_1:
    if current_love_stat + 5 > max_love_stat:
        $ current_love_stat += max_love_stat - current_love_stat
    elif True:
        $ current_love_stat += 5
    return

label exp_hobby_1:
    $ renpy.random.shuffle(store.hobby_uniclub)
    $ store.in_hobby_list = store.hobby_uniclub[0:3]
    $ renpy.show_screen("hobby_store", None, 0)
    return

label exp_hobby_2:
    $ renpy.random.shuffle(store.hobbypark_list)
    $ store.in_hobby_list = store.hobbypark_list[0:3]
    $ renpy.show_screen("hobby_store", None, 0)
    return

label exp_store_1:
    $ clothing_to_choose = [x for x in all_clothes_list if x.rarity == 1]
    $ renpy.random.shuffle(clothing_to_choose)
    $ store.in_store_list = clothing_to_choose[0:3]
    $ renpy.show_screen("clothes_store", shop=None, expshop=basic_store)
    return

label exp_new_card_1:
    $ cards_to_choose = [x for x in hand_lv5 if x not in starter_deck]
    $ renpy.random.shuffle(cards_to_choose)
    call screen get_a_card(cards_to_choose)
    return

label exp_remove_card_1:
    $ nr_row = math.ceil(len(store.starter_deck)/3)
    call screen remove_a_card(nr_row, store.starter_deck)
    return

label exp_sell_1:
    call screen trade_clothes
    return

label exp_upgrade_1:
    call upgrade_card_label from _call_upgrade_card_label_1
    return


label reward_restart:
    $ chosen_exp = "none_0"
    $ transition_clean_up()
    return

label reward_exp_1:
    $ exp_shop = [exp_basic_sell, exp_basic_present, exp_basic_hobby]
    $ renpy.random.shuffle(exp_shop)
    call screen exp_reward(exp_shop[0:3])
    call expression "exp_" + chosen_exp from _call_expression_6
    call reward_restart from _call_reward_restart
    return

label reward_exp_2:
    $ exp_shop = [exp_basic_present, exp_basic_hobby, exp_basic_newcard, exp_basic_shop, exp_remove_card]
    $ renpy.random.shuffle(exp_shop)
    call screen exp_reward(exp_shop[0:3])
    call expression "exp_" + chosen_exp from _call_expression_7
    call reward_restart from _call_reward_restart_1
    return

label reward_exp_3:
    $ exp_shop = [exp_basic_present, exp_basic_hobby, exp_basic_newcard, exp_basic_shop, exp_basic_sell, exp_upgrade_card, exp_remove_card]
    $ renpy.random.shuffle(exp_shop)
    call screen exp_reward(exp_shop[0:3])
    call expression "exp_" + chosen_exp from _call_expression_8
    call reward_restart from _call_reward_restart_2
    return

label reward_exp_4:
    $ exp_shop = [exp_basic_present, exp_basic_hobby, exp_basic_newcard, exp_basic_shop, exp_basic_sell, exp_upgrade_card, exp_remove_card]
    $ renpy.random.shuffle(exp_shop)
    call screen exp_reward(exp_shop[0:3])
    call expression "exp_" + chosen_exp from _call_expression_9
    call reward_restart from _call_reward_restart_3
    return

label reward_exp_5:
    $ exp_shop = [exp_basic_present, exp_basic_hobby_2, exp_basic_newcard, exp_basic_shop, exp_basic_sell, exp_upgrade_card, exp_remove_card]
    $ renpy.random.shuffle(exp_shop)
    call screen exp_reward(exp_shop[0:3])
    call expression "exp_" + chosen_exp from _call_expression_10
    call reward_restart from _call_reward_restart_4
    return

label reward_exp_6:
    $ exp_shop = [exp_basic_present, exp_basic_hobby_2, exp_basic_newcard, exp_basic_shop, exp_basic_sell, exp_upgrade_card, exp_remove_card]
    $ renpy.random.shuffle(exp_shop)
    call screen exp_reward(exp_shop[0:3])
    call expression "exp_" + chosen_exp from _call_expression_11
    call reward_restart from _call_reward_restart_5
    return






label d_lv1_1_Miriam:
    voice "voices/mc/mc_lv1_1a.ogg"
    mc "*sigh*... That's [persistent.selected_girl]... the most beautiful girl in the whole University..."
    voice "voices/mc/mc_lv1_1b.ogg"
    mc "Isn't she pretty?"
    voice "voices/mc/mc_lv1_1c.ogg"
    mc "The only problem is... she doesn't even know who I am..."
    return

label d_lv1_2_Miriam:
    mc "*sigh*... [persistent.selected_girl], [persistent.selected_girl], [persistent.selected_girl]..."
    voice "voices/mc/mc_lv1_2a.ogg"
    mc "She's the prettiest girl in the whole University."
    voice "voices/mc/mc_lv1_2b.ogg"
    mc "If only she knew me..."
    return

label d_lv1_3_Miriam:
    voice "voices/mc/mc_lv1_3a.ogg"
    mc "*sigh*... I've such a big crush on her..."
    voice "voices/mc/mc_lv1_3b.ogg"
    mc "[persistent.selected_girl]... the hottest girl on Campus."
    voice "voices/mc/mc_lv1_3c.ogg"
    mc "I just need to make her notice me..."
    return

label d_lv1_4_Miriam:
    voice "voices/mc/mc_lv1_4a.ogg"
    mc "*sigh* [persistent.selected_girl]... her golden hair and blue eyes are so beautiful..."
    voice "voices/mc/mc_lv1_4b.ogg"
    mc "No one else on campus compares to her..."
    voice "voices/mc/mc_lv1_4c.ogg"
    mc "Shame she doesn't even know me..."
    return


label d_lv2_1_Miriam:
    voice "voices/mc/mc_lv2_1a.ogg"
    mc "Look at her... she's always surrounded by guys..."
    voice "voices/mc/mc_lv2_1b.ogg"
    mc "I need to make a move if I don't want to lose her."
    return

label d_lv2_2_Miriam:
    voice "voices/mc/mc_lv2_2a.ogg"
    mc "She's so popular among guys..."
    voice "voices/mc/mc_lv2_2b.ogg"
    mc "I really need to get to know her before it's too late..."
    return

label d_lv2_3_Miriam:
    voice "voices/mc/mc_lv2_3a.ogg"
    mc "My God... there are so many guys chasing after her..."
    voice "voices/mc/mc_lv2_3b.ogg"
    mc "I need to talk to her before it's too late..."
    return

label d_lv2_4_Miriam:
    voice "voices/mc/mc_lv2_4a.ogg"
    mc "I get so jealous when I see her talking to other dudes..."
    voice "voices/mc/mc_lv2_4b.ogg"
    mc "I can't let one of them get to her first..."
    return


label d_lv3_1_Miriam:
    voice "voices/mc/mc_lv3_1a.ogg"
    mc "I can't believe we're friends now!"
    voice "voices/mc/mc_lv3_1b.ogg"
    mc "I need to keep talking to her!"
    return

label d_lv3_2_Miriam:
    voice "voices/mc/mc_lv3_2a.ogg"
    mc "That was easy!"
    voice "voices/mc/mc_lv3_2b.ogg"
    mc "Now I need to get to know her better."
    return

label d_lv3_3_Miriam:
    voice "voices/mc/mc_lv3_3a.ogg"
    mc "She was really nice!"
    voice "voices/mc/mc_lv3_3b.ogg"
    mc "I can't wait to spend more time with her."
    return

label d_lv3_4_Miriam:
    voice "voices/mc/mc_lv3_4a.ogg"
    mc "Oh boy, she's even cuter than I thought!"
    voice "voices/mc/mc_lv3_4b.ogg"
    mc "I really want to get to know her better."
    return


label d_lv4_1_Miriam:
    voice "voices/mc/mc_lv4_1a.ogg"
    mc "Things are going really well! I think I'll start flirting with her..."
    voice "voices/mc/mc_lv4_1b.ogg"
    mc "But what about all those other girls? She seemed to be getting jealous..."
    return

label d_lv4_2_Miriam:
    voice "voices/mc/mc_lv4_2a.ogg"
    mc "Things couldn't be better! Should I start flirting with her?"
    voice "voices/mc/mc_lv4_2b.ogg"
    mc "But I need to be careful with all the other cute girls..."
    voice "voices/mc/mc_lv4_2c.ogg"
    mc "[persistent.selected_girl] might not like if she sees me flirting with anyone else..."
    return

label d_lv4_3_Miriam:
    voice "voices/mc/mc_lv4_3a.ogg"
    mc "Gosh, I love spending time with [persistent.selected_girl]... I might even start flirting with her..."
    voice "voices/mc/mc_lv4_3b.ogg"
    mc "But... was she getting jealous of the other girls...?"
    return

label d_lv4_4_Miriam:
    voice "voices/mc/mc_lv4_4a.ogg"
    mc "She's so funny, I love hanging out with her!"
    voice "voices/mc/mc_lv4_4b.ogg"
    mc "But she seemed a bit jealous of other girls..."
    return



label d_lv5_1_Miriam:
    voice "voices/mc/mc_lv5_1a.ogg"
    mc "Hey, I was wondering... do you wanna go out do something together?"
    voice "voices/" + persistent.selected_girl + "/mer_lv5_1.ogg"
    g "Sure! I really like to run, do you wanna join me next time?"
    voice "voices/mc/mc_lv5_1b.ogg"
    mc "Absolutely!"
    return

label d_lv5_2_Miriam:
    voice "voices/mc/mc_lv5_2a.ogg"
    mc "Hey, I think it'd be fun to go do some activity together, what do you think?"
    voice "voices/" + persistent.selected_girl + "/mer_lv5_2.ogg"
    g "Sounds fun! What about going for a run in the park?"
    voice "voices/mc/mc_lv5_2b.ogg"
    mc "Sure, let's do it!"
    return

label d_lv5_3_Miriam:
    voice "voices/" + persistent.selected_girl + "/mer_lv5_3a.ogg"
    g "I really like to go for a run in the park. Do you wanna join me next time?"
    voice "voices/mc/mc_lv5_3a.ogg"
    mc "Definitely!"
    voice "voices/" + persistent.selected_girl + "/mer_lv5_3b.ogg"
    g "Haha, great! It'll be fun!"
    return

label d_lv5_4_Miriam:
    voice "voices/" + persistent.selected_girl + "/mer_lv5_4a.ogg"
    g "Do you wanna do some activity together one of these days?"
    voice "voices/mc/mc_lv5_4a.ogg"
    mc "Yes! What about going for a walk in the park?"
    voice "voices/" + persistent.selected_girl + "/mer_lv5_4b.ogg"
    g "Oh, I love the park! I go there running every week. Let's go together!"
    return


label d_lv6_1_Miriam:
    voice "voices/" + persistent.selected_girl + "/mer_lv6_1.ogg"
    g "This was really fun! Do you wanna rest?"
    voice "voices/mc/mc_lv6_1a.ogg"
    mc "Yeah. Let's just sit down and chat a little bit."
    voice "voices/mc/mc_lv6_1b.ogg"
    mc "(Man... I really need to tell her I like her...)"
    return

label d_lv6_2_Miriam:
    voice "voices/" + persistent.selected_girl + "/mer_lv6_2.ogg"
    g "Wow, that was fun! Are you tired?"
    voice "voices/mc/mc_lv6_2a.ogg"
    mc "Yeah, a little bit. Let's just sit down and chat for a bit."
    voice "voices/mc/mc_lv6_1b.ogg"
    mc "(Man... I really need to tell her I like her...)"
    return

label d_lv6_3_Miriam:
    voice "voices/" + persistent.selected_girl + "/mer_lv6_3.ogg"
    g "That was fun, but I am so tired now!"
    voice "voices/mc/mc_lv6_3a.ogg"
    mc "Haha, that's normal. Let's just rest a little bit."
    voice "voices/mc/mc_lv6_1b.ogg"
    mc "(Man... I really need to tell her I like her...)"
    return

label d_lv6_4_Miriam:
    voice "voices/" + persistent.selected_girl + "/mer_lv6_4.ogg"
    g "Oof, I can't do it anymore! I'm done!"
    voice "voices/mc/mc_lv6_4a.ogg"
    mc "Haha, it was tough! Let's slow down and just chat for a bit."
    voice "voices/mc/mc_lv6_1b.ogg"
    mc "(Man... I really need to tell her I like her...)"
    return





label change_lvl_2_Miriam:
    $ start_change_girl_image_logic("lv2b")
    show screen player_interface with Dissolve (transition_time)
    voice "voices/mc/mc_change2.ogg"
    mc "H-hi... nice to meet you."
    voice "voices/" + persistent.selected_girl + "/mer_change2.ogg"
    g "Hi, there! We have some classes together, don't we?"
    call end_change_girl_image_logic_label from _call_end_change_girl_image_logic_label
    return

label change_lvl_3_Miriam:
    $ start_change_girl_image_logic("lv3b")
    show screen player_interface with Dissolve (transition_time)
    voice "voices/" + persistent.selected_girl + "/mer_change3.ogg"
    g "Haha, you're really funny! I like you..."
    voice "voices/mc/mc_change3.ogg"
    mc "*whelp* Thank you!"
    call end_change_girl_image_logic_label from _call_end_change_girl_image_logic_label_1
    return

label change_lvl_4_Miriam:
    $ start_change_girl_image_logic("lv4b")
    show screen player_interface with Dissolve (transition_time)
    voice "voices/mc/mc_change4.ogg"
    mc "Oh wow... those shorts are capable of breaking a few hearts if you aren't careful..."
    voice "voices/" + persistent.selected_girl + "/mer_change4.ogg"
    g "Haha! You really think they look that good on me? Thank you!"
    call end_change_girl_image_logic_label from _call_end_change_girl_image_logic_label_2
    return

label change_lvl_5_Miriam:
    $ start_change_girl_image_logic("lv5b")
    show screen player_interface with Dissolve (transition_time)
    voice "voices/" + persistent.selected_girl + "/mer_change5.ogg"
    g "We need to stretch if we don't want to get sore muscles! Are you ready?"
    voice "voices/mc/mc_change5.ogg"
    mc "Yeah, you're absolutely right!"
    call end_change_girl_image_logic_label from _call_end_change_girl_image_logic_label_3
    return

label change_lvl_6_Miriam:
    $ start_change_girl_image_logic("lv6b")
    show screen player_interface with Dissolve (transition_time)
    voice "voices/mc/mc_change6a.ogg"
    mc "[persistent.selected_girl], there's something I need to tell you... I really like you..."
    voice "voices/" + persistent.selected_girl + "/mer_change6.ogg"
    g "Oh. You... like me?"
    voice "voices/mc/mc_change6b.ogg"
    mc "I do..."
    call end_change_girl_image_logic_label from _call_end_change_girl_image_logic_label_4
    return